Maple-assisted proof of empirical formula for A183326

Robert Israel

1 May 2019

There are $2^{10} = 1024$ possible configurations for two adjacent rows, but only those where each 1 is adjacent to one or two other 1's is allowed. The following code computes the allowed configurations, of which there turn out to be 280.

```
> Configs:= NULL: nConfigs:= 0:
  for n from 0 to 1023 do
    v:= convert(1024+n,base,2)[1..10];
    good:= true:
    for i from 1 to 10 do
      if v[i] = 1 then
        t:= 0:
        if i mod 5 <> 1 then t:= t + v[i-1] fi;
        if i mod 5 <> 0 then t:= t + v[i+1] fi;
        if i <= 5 then t:= t + v[i+5] else t:= t + v[i-5] fi;
        if t = 0 or t > 2 then good:= false; break fi
      fi
    od;
    if good then Configs:= Configs, v; nConfigs:= nConfigs+1; fi
  od:
  Configs:= [Configs]:
  nConfigs;
```

$$280 \quad \textbf{(1)}$$

Let T be the 280×280 transition matrix such that $T_{ij} = 1$ if the bottom two rows of a 3×5 array could be in configuration i while the top two rows are in configuration j (thus the middle row is both the top row of configuration i and the bottom row of configuration j, and every 1 in the middle row has exactly two neighbours that are 1's). The following code computes it.

```
> q:= proc(i,j) local Ci, Cj,k;
     Ci:= Configs[i]; Cj:= Configs[j];
     if Ci[1..5] <> Cj[6..10] then return 0 fi;
     if Ci[1] = 1 and Ci[6]+Ci[2]+Cj[1] <> 2 then return 0 fi;
     if Ci[5] = 1 and Ci[4]+Ci[10]+Cj[5] <> 2 then return 0 fi;
     for k from 2 to 4 do
       if Ci[k] = 1 and Ci[k-1]+Ci[k+1]+Ci[k+5]+Cj[k] <> 2 then
  return 0 fi
     od;
     1
  end proc:
  T:= Matrix(280,280, q):
```

Thus $a(n) = u^T T^{n+2} u$ where u is a column vector with $u_1 = 1$ (corresponding to the configuration of all 0's, which is the first allowed configuration).

```
> u:= Vector([1,0$279]):
```

To check, here are the first few entries of our sequence. For future use, I compute $V_n = T^n u$ for some values of n.

```
> V[0]:= u:
  for n from 1 to 26 do V[n]:= T . V[n-1] od:
> seq(u^%T . V[n], n = 3 .. 26);
```

1, 6, 19, 72, 289, 996, 3325, 11415, 39720, 138689, 483837, 1682961, 5845649, 20310166, 70604782, 245504404, 853649448, 2967979455, 10318546476, 35873587105, 124720541039, 433616480871, 1507558685202, 5241330944265 **(2)**

Now here is the empirical recurrence formula. It says that $u^T T^{n+2} Q(T)\, u = 0$ for all positive integers n, where Q is the following polynomial.

```
> n:= 'n': t:= 't':
  empirical:= a(n)=5*a(n-1)-8*a(n-2)+9*a(n-3)-2*a(n-4)+14*a(n-5)+3*
  a(n-6)-44*a(n-7)+18*a(n-8)+29*a(n-9)-10*a(n-10)-69*a(n-11)+16*a
  (n-12)+87*a(n-13)+15*a(n-14)-55*a(n-15)-40*a(n-16)+6*a(n-17)+9*a
  (n-18)+4*a(n-19)-2*a(n-20):
  Q:= unapply(add(coeff((lhs-rhs)(empirical),a(n-i))*t^(20-i),i=0.
  .20),t);
```

$$Q := t \mapsto t^{20} - 5\,t^{19} + 8\,t^{18} - 9\,t^{17} + 2\,t^{16} - 14\,t^{15} - 3\,t^{14} + 44\,t^{13} - 18\,t^{12} - 29\,t^{11} + 10\,t^{10} + 69\,t^{9} - 16\,t^{8} - 87\,t^{7} - 15\,t^{6} + 55\,t^{5} + 40\,t^{4} - 6\,t^{3} - 9\,t^{2} - 4\,t + 2 \tag{3}$$

In fact, it turns out that $T\,Q(T)\,u = 0$. Verifying this completes the proof.

```
> Qu:= add(coeff(Q(t),t,j)*V[j],j=0..20):
> TQu:= T . Qu:
  TQu^%T . TQu;
```

$$0 \tag{4}$$